
Παράδειγμα 1 

Εύρεση της τριάδας ακεραίων αριθμών X,Y,Z με Χ  [4,8],  Υ  [1,8], Ζ  [6,7,8,9,10] και το Χ είναι κατά 1 μεγαλύτερο 

του Υ και το Ζ είναι μικρότερο του Χ. 

Με χρήση της απλής οπισθοδρόμησης της Prolog  

solve(X,Y,Z) :- 
             member(X,[4,5,6,7,8]), 
             member(Y,[1,2,3,4,5,6,7,8]), 
             member(Z,[6,7,8,9,10]), 
             X =:= Y +1, 
             Z<X, 
             write([X,Y,Z]),nl,fail. 

lab05_01a.pl 

 

?- consult(‘lab05_01a’). 

?- solve(X,Y,Z). 
[7,6,6] 
[8,7,6] 
[8,7,7] 
false. 

 

Με χρήση της βιβλιοθήκης CLPFD (Constraint Logic Programming over Finite Domains). 

:- use_module(library(clpfd)). 
 
solve(X,Y,Z):- 
              X in 4..8, 
              Y in 1..8, 
              Z in 6..10, 
              X #= Y+1, 
              Z #< X, 
              label([X,Y,Z]), 
              write([X,Y,Z]),nl,fail. 

lab05_01b.pl 

 

?- consult(‘lab05_01b’). 

?- solve(X,Y,Z). 
[7,6,6] 
[8,7,6] 
[8,7,7] 
false. 

 

Παράδειγμα 2 
Περιγραφή ως πρόβλημα CLP (λογικού προγραμματισμού με περιορισμούς) των προβλημάτων:  

α) το Χ είναι ακέραιος από 0 έως και 10, το Υ είναι ακέραιος από 4 έως και 8 και το Χ είναι μεγαλύτερο του Υ. 
β) το Χ είναι ακέραιος από 0 έως και 5, το Υ είναι ακέραιος από 4 έως και 8 και το Χ είναι μεγαλύτερο του Υ. 

:- use_module(library(clpfd)). 
 
test1(X,Y):- 
           X in 0..10, 
           Y in 4..8, 
           X #> Y. 
 
test1 :- 
      test1(X,Y), 
      label([X,Y]), 
      write([X,Y]),nl,fail. 
 
test2(X,Y):- 
           X in 0..5, 
           Y in 4..8, 
           X #> Y. 

lab05_02.pl 

 

?- consult(‘lab05_02’). 

?- test1(X,Y). 
X in 5..10, 
Y#=<X+ -1, 
Y in 4..8. 
 
?- test1. 
[5,4] 
[6,4] 
[6,5] 
[7,4] 
[7,5] 
[7,6] 
[8,4] 
[8,5] 
[8,6] 
[8,7] 
[9,4] 
[9,5] 
[9,6] 
[9,7] 
[9,8] 
[10,4] 
[10,5] 
[10,6] 
[10,7] 
[10,8] 
false. 
 
?- test2(X,Y). 
X = 5, 
Y = 4. 

Παράδειγμα 3 (alldifferent) 
Επίλυση του προβλήματος SEND+MORE=MONEY όπου κάθε χαρακτήρας είναι ένας ακέραιος αριθμός από το 0 μέχρι 
και το 9 (το S και το M δεν μπορούν να είναι μηδέν). 

:- use_module(library(clpfd)). 
 
puzzle([S,E,N,D] + [M,O,R,E] = [M,O,N,E,Y]) :- 
        Vars = [S,E,N,D,M,O,R,Y], 
        Vars ins 0..9, 
        all_different(Vars), 
                  S*1000 + E*100 + N*10 + D + 
                  M*1000 + O*100 + R*10 + E #= 
        M*10000 + O*1000 + N*100 + E*10 + Y, 
        M #\= 0, S #\= 0, 
        label(Vars). 

lab05_03.pl 

 

?- consult(‘lab05_03’). 

?- puzzle(X). 
X =  ([9, 5, 6, 7]+[1, 0, 8, 5]=[1, 0, 6, 5, 2]) ; 
false. 

 

Παράδειγμα 4 
Χρωματισμός χάρτη Αυστραλίας χρησιμοποιώντας διαφορετικά χρώματα για περιοχές που είναι γειτονικές. 

:- use_module(library(clpfd)). 
 
colourable([WA, SA, NT, Q, NSW, V, T]) :- 
                rgb(WA), rgb(SA), rgb(NT), rgb(Q), rgb(NSW), rgb(V), rgb(T), 
                WA #\= NT, NT #\= Q, Q #\= NSW, NSW #\= V, 
                SA #\= WA, SA #\= NT, SA #\= Q, SA #\= NSW, SA #\= V. 
 
rgb(1). 
rgb(2). 
rgb(3). 

lab05_04.pl 

 

?- consult(‘lab05_04’). 

?- colourable(X). 
X = [1, 2, 3, 1, 3, 1, 1] ; 
X = [1, 2, 3, 1, 3, 1, 2] ; 
X = [1, 2, 3, 1, 3, 1, 3] ; 
X = [1, 3, 2, 1, 2, 1, 1] ; 
X = [1, 3, 2, 1, 2, 1, 2] ; 
X = [1, 3, 2, 1, 2, 1, 3] ; 
X = [2, 1, 3, 2, 3, 2, 1] ; 
X = [2, 1, 3, 2, 3, 2, 2] ; 
X = [2, 1, 3, 2, 3, 2, 3] ; 
X = [2, 3, 1, 2, 1, 2, 1] ; 
X = [2, 3, 1, 2, 1, 2, 2] ; 
X = [2, 3, 1, 2, 1, 2, 3] ; 
X = [3, 1, 2, 3, 2, 3, 1] ; 
X = [3, 1, 2, 3, 2, 3, 2] ; 
X = [3, 1, 2, 3, 2, 3, 3] ; 
X = [3, 2, 1, 3, 1, 3, 1] ; 

X = [3, 2, 1, 3, 1, 3, 2] ; 
X = [3, 2, 1, 3, 1, 3, 3] ; 
false. 

 

Παράδειγμα 5 (labelling) 
Η συνολική τιμή 4 αντικειμένων είναι 7,11€. Το γινόμενο των τιμών τους είναι επίσης 7,11€. Ποιες είναι οι τιμές των 
4 αντικειμένων; 

:- use_module(library(clpfd)). 
 
solve(Vs) :- Vs = [A,B,C,D], 
Vs ins 0..711, 
A*B*C*D #= 711*100^3, 
A+B+C+D #= 711, 
A#>=B, B#>=C, C#>=D, 
labeling([ff],Vs). 

triska_example2_1.pl 

 

?- consult(‘lab05_05’). 

?- solve(X). 
X = [316, 150, 125, 120] ; 
false. 

 

Παράδειγμα 6 (length) 
Να βρεθούν όλοι οι συνδυασμοί 4 ακεραίων που ανήκουν είτε στο διάστημα 1 έως και 3 είτε στο 7 έως και 9, που 
είναι διαφορετικοί μεταξύ τους και η πρώτη τιμή είναι μεγαλύτερη της τελευταίας. 

:- use_module(library(clpfd)). 
 
find4(Vs) :- 
          length(Vs,4), 
          Vs ins 1..3 \/ 7..9, 
          all_different(Vs), 
          Vs = [X,_,_,Y], 
          X #> Y, 
          label(Vs), 
          write(Vs),nl, 
          fail. 

lab05_06a.pl 

 

?- consult(‘lab05_06a’) 

?- find4(Vs). 
[2,3,7,1] 
[2,3,8,1]  
… 
[9,8,7,2] 
[9,8,7,3] 
false. 

